

The impact of misspecified covariate models on inclusion and omission bias when using fixed effects and full random effects models

Joakim Nyberg¹ and E. Niclas Jonsson¹

¹ Pharmetheus AB, Sweden

Corresponding Author:

Joakim Nyberg

Pharmetheus AB,

Kungsängstull 4

753 19 Uppsala, Sweden

email: joakim.nyberg@pharmetheus.com

Tel: +46(0)7 01 730 963

Model 1 - Simulation model: FEM, with covariate WT on both CL and V parameters; with IIV correlation between CL and V

```

$PROBLEM      TRUE FEM sim model cov on both parameters, Correlation
$INPUT        ID TIME DV EVID AMT WT LWT
$DATA         input_ffem.csv IGNORE=@
$SUBROUTINE ADVAN2 TRANS2
$PK

      TVKA  = THETA(1)
      TVCL  = THETA(2)
      TVV   = THETA(3)

      MU_1  = LOG(TVKA)
      MU_2  = LOG(TVCL)
      MU_3  = LOG(TVV)

      COVCL = 0
      COVV  = 0

      ALLCL = THETA(4)
      ALLV  = THETA(5)
      COVCL = ALLCL*LOG(WT/70)
      COVV  = ALLV*LOG(WT/70)

      KA = EXP(MU_1 + ETA(1))
      CL = EXP(MU_2 + COVCL + ETA(2))
      V  = EXP(MU_3 + COVV + ETA(3))

      S2   = V

$ERROR

      CP    = A(2)/V
      Y     = CP * EXP(EPS(1))

$THETA (0,0.66) ; 1. TVKA
$THETA (0,9)    ; 2. TVCL
$THETA (0,100)  ; 3. TVV
$THETA 0.75     ; 4. ALLO CL
$THETA 1        ; 5. ALLO V
$OMEGA 0.25     ; 1. OMEGA_KA
$OMEGA BLOCK(2)
      0.09      ; 2. OMEGA_CL
      0.036 0.09 ; OMEGA_CL_V 3. OMEGA_V
$SIGMA 0.0225   ; Res error
$SIMULATION (652911) ONLYSIM
$TABLE      ID TIME DV EVID AMT WT LWT NOPRINT NOAPPEND ONEHEADER
            FILE=mc-sim-1.dat

```

Model 2 - Estimation model: FEM, with covariate WT on both CL and V parameters; with IIV correlation between CL and V; estimated covariate coefficients

```
$PROB FEM model Cov on Both parameters, Corr, estimate covariates
$INPUT ID TIME DV EVID AMT WT LWT
$DATA input_ffem.csv IGNORE=@
$SUBROUTINES ADVAN2 TRANS2
$PK

      TVKA  = THETA(1)
      TVCL  = THETA(2)
      TVV   = THETA(3)

      MU_1  = LOG(TVKA)
      MU_2  = LOG(TVCL)
      MU_3  = LOG(TVV)

      COVCL = 0
      COVV  = 0

      ALLCL = THETA(4)
      ALLV  = THETA(5)
      COVCL = ALLCL*LOG(WT/70)
      COVV  = ALLV*LOG(WT/70)

      KA = EXP(MU_1 + ETA(1))
      CL = EXP(MU_2 + COVCL + ETA(2))
      V  = EXP(MU_3 + COVV + ETA(3))

      S2   = V

$ERROR

      CP    = A(2)/V
      Y     = CP * EXP(EPS(1))

$THETA (0,0.66) ; 1. TVKA
$THETA (0,9)    ; 2. TVCL
$THETA (0,100)  ; 3. TVV
$THETA 0.75     ; 4. WT on CL
$THETA 1        ; 5. WT on V

$OMEGA 0.25; 1. OMEGA_KA
$OMEGA BLOCK(2) 0.09; 2. OMEGA_CL
      0.036 0.09; OMEGA_CL_V 3. OMEGA_V

$SIGMA 0.0225; Res error number 1
$ESTIMATION METHOD=IMPMAP AUTO=1 RANMETHOD=3P INTER NOABORT PRINT=1
      CITER=15 NOCOV=1 ISAMPEND=100 NITER=300 ISAMPLE=300 CTYPE=1
      CALPHA=0.001
$ESTIMATION METHOD=IMPMAP INTER EONLY=1 NITER=5 ISAMPLE=1000 NOCOV=0
$COVARIANCE PRINT=E UNCONDITIONAL
```

**Model 3 - Estimation model: FREM, with covariate WT on all CL, V, and Ka parameters;
with IIV correlation between CL, V, and Ka; estimated covariate coefficients**

```
$PROBLEM      FREM FULL BLOCK
$INPUT        ID TIME DV EVID AMT WT LWT FREMTYPE
$DATA         mc-sim-FREM-1.dat IGNORE=@
$SUBROUTINE   ADVAN2 TRANS2
$PK

TVKA  = THETA(1)
TVCL  = THETA(2)
TVV   = THETA(3)

MU_1  = LOG(TVKA)
MU_2  = LOG(TVCL)
MU_3  = LOG(TVV)

KA    = EXP(MU_1 + ETA(1))
CL    = EXP(MU_2 + ETA(2))
V     = EXP(MU_3 + ETA(3))

S2    = V

MU_4  = THETA(4)

COV4  = MU_4 + ETA(4)

$ERROR
CP     = A(2)/V
Y      = CP * EXP(EPS(1))

;;;FREM CODE BEGIN COMPACT
      IF (FREMTYPE.EQ.100) THEN
        Y = COV4 + EPS(2)
        IPRED = COV4
      END IF
;;;FREM CODE END COMPACT

$THETA  (0,0.697851) ; 1. TVKA
$THETA  (0,10.324)   ; 2. TVCL
$THETA  (0,97.7813)  ; 3. TVV
$THETA  4.24849524204936 FIX ; TV_LWT
$OMEGA  BLOCK(4)
  0.0586899          ; 1. IIV on KA
  0.01 0.127874      ; 2. IIV on CL
  0.01 0.01 0.0925178 ; 3. IIV on V
  0.01 0.1 0.1 0.5   ;   BSV_LWT
$SIGMA  0.0217889    ;   1. RUW
$SIGMA  1E-07 FIX    ;   EPSCOV
$ESTIMATION METHOD=IMPMAP AUTO=1 RANMETHOD=3P INTER NOABORT PRINT=1
          CITER=15 NOCOV=1 ISAMPEND=100 NITER=300 ISAMPLE=300
          CTYPE=1 CALPHA=0.001
$ESTIMATION METHOD=IMPMAP INTER EONLY=1 NITER=5 ISAMPLE=1000 NOCOV=0
$COVARIANCE PRINT=E UNCONDITIONAL
```